Några MACRO's för IDPAC

Gustavsson, Ivar

1979

Document Version:
Förlagets slutgiltiga version

CODEN: LUTFD2/(TFRT-7170)/1-39/(1979)

NÅGRA MACRO'S FÖR IDPAC

I GUSTAVSSON

DEPARTMENT OF AUTOMATIC CONTROL
LUND INSTITUTE OF TECHNOLOGY
AUGUST 1979

NÄGRA MACRO´S FÖR IDPAC

I Gustavsson

Detta arbete har understötts av Styrelsen för Teknisk Utveckling under anslag nr 78-3763.

Dokumentutgivare
04T0 Lund Institute of Technology
Handläggare Department of Automatic Control
06T0 I Gustavsson
Författare
08T0 I Gustavsson

Dokumentnamn
04T6 REPORT
Utgivningsdatum
06T4 August 1979

Dokumentbeteckning
04T6 LUTFD2/(TFRT-7170)/1-39/(1979)
Ärendebeteckning
06T6

10T4

Dokumenttitel och undertitel
18T0
Några MACRO´s för IDPAC

(Some MACRO´s for IDPAC)

Referat (sammandrag)
26T0
A number of MACRO´s for use in the interactive identification program package IDPAC are presented in this report. They concern the areas of correlation and spectral analysis, least squares identification and model analysis. Some of them are primarily designed for teaching purposes, but most of them are suitable for practical use.

Referat skrivet av
42T0 Author

Förslag till ytterligare nyckelord
44T0

Klassifikationssystem och -klass(er)
50T0

Indextermer (ange källa)
52T0

Omfång
54T0 39 pages

Övriga bibliografiska uppgifter
56T2

Språk
58T0 Swedish

Sekretessuppgifter
60T0

ISSN
60T4

ISBN
60T6

Dokumentet kan erhållas från
62T0
Department of Automatic Control
Lund Institute of Technology
Box 725, S-220 07 LUND 7, Sweden

Mottagarens uppgifter
62T4

Pris
66T0

SIS-DB 1 DOKUMENTDATABLAD enligt SIS 62 10 12

INNEHÅLLSFÖRTECKNING

INLEDNING

Denna rapport innehåller ett antal MACRO´s, varav en del, typ CORANA, i första hand är avsedda för undervisningsändamål, medan de flesta är avsedda för praktiska sammanhang. Avsikten är dels att publicera ett antal ofta efterfrågade MACRO´s, typ LSID, som underlättar minsta kvadratidentifiering, men dels också ge exempel på hur användaren själv kan skriva sina egna speciella MACRO´s och därvid utnyttja IDPAC effektivare.

Före varje MACRO ges en mycket kort beskrivning och sedan listas själva MACRO´n med sina kommentarer. Det rekommenderas att användaren läser såväl den inledande beskrivningen som kommentarerna i MACRO´n före användandet.

De MACRO´s som presenteras kan delas in i grupper enligt följande:

Korrelations- och spektralanalys

CORANA	Fråge/svarsstyrt program för korrelations- och spektralanalys
PREWHITE	Förfiltrering (prewhitening)
CCOFF	Beräkning och plottning av korskorrelationsfunktion
LINSPEC	Beräkning och plottning av autospektrum (beräknat för linjärt ekvidistanta punkter)
INTSPEC	Beräkning och plottning av integrerat spectrum
TRF	Skattning och plottning av överföringsfunktion
COH	Beräkning och plottning av koherensfunktion
DFTT	Beräkning, utjämning och plottning av autospektrum (DFT) (utjämning genom uppdelning av tidsserien i mindre block och medelvärdesbildning över de skattade spektra)
DFTF	Beräkning, utjämning och plottning av autospektrum (DFT) (utjämning genom medelvärdesbildning i frekvensplanet)

Minsta kvadratidentifiering

LSID	Minsta kvadratidentifiering av modell med givet ordningstal
LSORD	Bestämning av ordningstal vid minsta kvadratidentifiering
LSDEL	Bestämning av den rena tidsfördröjningen för en process

Modellanalys

FTST	Beräkning av testkvantiteten för F-test
MODANA	Modellanalys med test av residualer, beräkning av modellens deterministiska utsignal och dess överföringsfunktion
STEPTEST	Beräkning och plottning av ett systems stegsvar
IMPTEST	Beräkning och plottning av ett systems impulssvar
RAND	Test av en modells osäkerhet genom en Monte Carlo simulering av modellens tidssvar på en given insignal
PREDERR	Beräkning och plottning av k-stegs prediktionsfelet för ett givet system

CORANA

Denna MACRO realiserar ett fråge/svarsstyrt program för korrelations- och spektralanalys. De saker man kan göra i denna MACRO framgår av följande tabell.

Beräkning av autokorrelationsfunktion
Beräkning av autospektrum
Beräkning av korskorrelationsfunktion
Beräkning av korsspektrum
Skattning av överföringsfunktion
Beräkning av koherensfunktion
Beräkning av spektrum med DFT (Discrete Fourier Transform)
Skattning av överföringsfunktion med DFT
Förfiltrering (prewhitening)
Plottning av spektra och överföringsfunktioner
Plottning av tidsserier

Denna MACRO är avsedd i första hand för nybörjare i IDPAC-användning. En lämplig samling övningar med lösningar finns i Gustavsson och Nilsson (1979).

MACRO´n startas med kommandot

>CORANA

Vissa av denna MACRO´ns möjligheter har implementerats som separata MACRO´s för att öka effektiviteten. Det finns alltså

PREWHITE - Förfiltrering (prewhitening) av data (sid 14)
CCOFF - Beräkning och plottning av korskorrelationsfunktion (sid 16)
TRF - Skattning och plottning av överföringsfunktion (sid 20)
COH - Beräkning och plottning av koherensfunktion (sid 21)

Delen DFT har utvidgats och inkluderar möjligheter för utjämning (användning av fönster) i följande MACRO´s:

DFTT - Utjämning genom uppdelning av tidsserien i block (sid 22)
DFTF - Utjämning genom medelvärdesbildning i frekvensplanet (sid 24)

```
MACRO CORANA
"
"MACRO'S NEEDED
"       MACKA
"       KILL
"
"
"
WRITE (FF) ' '
WRITE 'THIS QUESTION AND ANSWER PROGRAM IS INTENDED TO GUIDE YOU'
WRITE 'THROUGH THE EXERCISE IN CORRELATION ANALYSIS'
WRITE ' '
WRITE ' '
WRITE ' '
WRITE 'HIT THE RETURN KEY TO PROCEED'
"
MACKA
"
LET I=UNASS.
READ ; I INT
"
"
"
LABEL MENU
WRITE (FF) ' '
WRITE 'CHOOSE BETWEEN THE FOLLOWING POSSIBILITIES'
WRITE ' '
WRITE 'ACOF     - COMPUTATION OF THE AUTOCORRELATION FUNCTION'
WRITE 'ASPEC    - COMPUTATION OF THE AUTOSPECTRUM'
WRITE 'CCOF     - COMPUTATION OF THE CROSSCORRELATION FUNCTION'
WRITE 'CSPEC    - COMPUTATION OF THE CROSSPECTRUM'
WRITE 'TRF      - ESTIMATION OF THE TRANSFER FUNCTION'
WRITE 'COH      - ESTIMATION OF THE COHERENCE FUNCTION'
WRITE 'DFT      - ESTIMATION OF SPECTRUM USING DFT'
WRITE 'TRFDFT   - ESTIMATION OF TRANSFER FUNCTION USING DFT'
WRITE 'PREWHITE- PREWHITENING OF DATA FOR CROSSCORRELATION ANALYSIS'
WRITE 'BODE     - PLOTTING OF SPECTRA AND TRANSFER FUNCTIONS'
WRITE 'PLOT     - PLOTTING OF TIME SERIES'
WRITE 'EXIT     - EXIT TO COMMAND MODE'
"
LABEL 30
WRITE ' '
WRITE 'WHAT IS YOUR INTEREST?'
READ ANSWER NAME
"
IF ANSWER EQ ACOF GOTO ACOF
IF ANSWER EQ ASPEC GOTO ASPEC
IF ANSWER EQ CCOF GOTO CCOF
IF ANSWER EQ CSPEC GOTO CSPEC
IF ANSWER EQ TRF GOTO TRF
IF ANSWER EQ COH GOTO COH
IF ANSWER EQ DFT GOTO DFT
IF ANSWER EQ TRFDFT GOTO TRFDFT
IF ANSWER EQ PREWHITE GOTO PREWHITE
IF ANSWER EQ BODE GOTO BODE
IF ANSWER EQ PLOT GOTO PLOT
IF ANSWER EQ EXIT GOTO END
GOTO EXIT
"
"
"
LABEL ACOF
"
WRITE 'NAME OF DATA FILE WHERE THE SIGNAL TO BE ANALYSED IS'
```

```
READ DATA NAME
"
WRITE 'IF DATA IN COLUMN 1 IS TO BE ANALYSED JUST HIT RETURN KEY'
WRITE 'OTHERWISE TYPE THE DESIRED COLUMN NUMBER'
LET N=UNASS.
READ ; N INT
DEFAULT N=1
"
WRITE 'CHOOSE NAME OF FILE FOR THE RESULTING CORRELATION FUNCTION'
READ ZQAC NAME
"
WRITE 'CHOOSE THE MAXIMUM NUMBER OF LAGS IN THE COMPUTATION OF'
WRITE 'THE AUTOCORRELATION FUNCTION'
READ NLAG INT
"
ACOF ZQAC<DATA(N) NLAG
"
PLOT (NLAG) ZQAC
"
GOTO EXIT
"
"
"
LABEL ASPEC
"
WRITE 'NAME OF DATA FILE WHERE THE SIGNAL TO BE ANALYSED IS'
READ DATA NAME
"
WRITE 'IF DATA IN COLUMN 1 IS TO BE ANALYSED JUST HIT THE RETURN KEY'
WRITE 'OTHERWISE TYPE THE DESIRED COLUMN NUMBER'
LET N=UNASS.
READ ; N INT
DEFAULT N=1
"
WRITE 'CHOOSE NAME OF FILE FOR THE ESTIMATED AUTOSPECTRUM'
READ ZQAS NAME
"
WRITE 'CHOOSE THE NUMBER OF LAGS IN THE COMPUTATION OF THE AUTO-'
WRITE 'CORRELATION FUNCTION, I.E. THE WIDTH OF THE TIME WINDOW'
READ NLAG INT
"
ASPEC ZQAS<DATA(N) NLAG
"
BODE ZQAS
"
GOTO EXIT
"
"
"
LABEL CCOF
"
WRITE 'NAME OF DATA FILE WHERE YOUR INPUT SIGNAL IS'
READ F1 NAME
"
WRITE 'IF DATA IN COLUMN 1 IS THE INPUT JUST HIT THE RETURN KEY'
WRITE 'OTHERWISE TYPE THE DESIRED COLUMN NUMBER'
LET N1=UNASS.
READ ; N1 INT
DEFAULT N1=1
"
WRITE 'NAME OF DATA FILE WHERE YOUR OUTPUT SIGNAL IS'
READ F2 NAME
"
WRITE 'IF DATA IN COLUMN 1 IS THE OUTPUT JUST HIT THE RETURN KEY'
```

```
WRITE 'OTHERWISE TYPE THE DESIRED COLUMN NUMBER'
LET N2=UNASS.
READ ; N2 INT
DEFAULT N2=1
"
WRITE 'CHOOSE NAME OF FILE FOR THE ESTIMATED CORRELATION FUNCTION'
READ ZQCC NAME
"
WRITE 'CHOOSE THE MAXIMUM NUMBER OF LAGS IN THE COMPUTATION OF'
WRITE 'THE CROSSCORRELATION FUNCTION'
READ NLAG INT
"
LET NS = 2 * NLAG
LET NN = NS + 1
LET NQ = NLAG + 1
"
INSI ZQP NN
LET AMP.=2.
LET IFP.=NQ
ZERO
STEP
LET AMP.=1.
LET IFP.=1
RAMP
X
"
SCLOP ZQP(2)<ZQP(2) - 1.
SCLOP ZQP(4)<ZQP(1) + 1.
SCLOP ZQP(3)<ZQP(3) - NLAG
"
CCOF ZQCC<F1(N1) F2(N2) NLAG
"
PLOT (NM) (NN) ZQP(3) < ZQCC ZQP(1 4) (HP) ZQP(2)
"
GOTO EXIT
"
"
"
LABEL CSPEC
"
WRITE 'NAME OF DATA FILE WHERE YOUR INPUT SIGNAL IS'
READ F1 NAME
"
WRITE 'IF DATA IN COLUMN 1 IS THE INPUT JUST HIT THE RETURN KEY'
WRITE 'OTHERWISE TYPE THE DESIRED COLUMN NUMBER'
LET N1=UNASS.
READ ; N1 INT
DEFAULT N1=1
"
WRITE 'NAME OF DATA FILE WHERE YOUR OUTPUT SIGNAL IS'
READ F2 NAME
"
WRITE 'IF DATA IN  COLUMN 1 IS THE OUTPUT JUST HIT THE RETURN KEY'
WRITE 'OTHERWISE TYPE THE DESIRED COLUMN NUMBER'
LET N2=UNASS.
READ ; N2 INT
DEFAULT N2=1
"
WRITE 'CHOOSE NAME OF FILE FOR THE ESTIMATED CROSSPECTRUM'
READ ZQCS NAME
"
WRITE 'CHOOSE THE NUMBER OF LAGS IN THE COMPUTATION OF THE CORRELATION'
WRITE 'FUNCTIONS, I.E. THE WIDTH OF THE TIME WINDOW'
READ NLAG INT
```

```
"
CSPEC ZQCS<F1(N1) F2(N2) NLAG
"
BODE ZQCS
"
GOTO EXIT
"
"
"
LABEL TRF
"
WRITE 'NAME OF DATA FILE WHERE YOUR INPUT SIGNAL IS'
READ F1 NAME
"
WRITE 'IF DATA IN COLUMN 1 IS THE INPUT JUST HIT THE RETURN KEY'
WRITE 'OTHERWISE TYPE THE DESIRED COLUMN NUMBER'
LET N1=UNASS.
READ ; N1 INT
DEFAULT N1=1
"
WRITE 'NAME OF DATA FILE WHERE YOUR OUTPUT IS'
READ F2 NAME
"
WRITE 'IF DATA IN COLUMN 1 IS THE OUTPUT JUST HIT THE RETURN KEY'
WRITE 'OTHERWISE TYPE THE DESIRED COLUMN NUMBER'
LET N2=UNASS.
READ ; N2 INT
DEFAULT N2=1
"
WRITE 'CHOOSE NAME OF FILE FOR THE ESTIMATED TRANSFER FUNCTION'
READ ZQTRF NAME
"
WRITE 'CHOOSE THE NUMBER OF LAGS IN THE COMPUTATION OF CORRELATION'
WRITE 'FUNCTIONS, I.E. THE WIDTH OF THE TIME WINDOW'
READ NLAG INT
"
SWITCH ECHO ON
"
ASPEC ZQAU<F1(N1) NLAG
CSPEC ZQCUY<F1(N1) F2(N2) NLAG
FROP ZQTRF<ZQCUY/ZQAU
"
SWITCH ECHO OFF
"
LET N3=UNASS.
WRITE 'HIT THE RETURN KEY TO PROCEED'
READ ; N3 INT
"
BODE ZQTRF
"
GOTO EXIT
"
"
"
LABEL COH
"
WRITE 'NAME OF DATA FILE WHERE YOUR INPUT SIGNAL IS'
READ F1 NAME
"
WRITE 'IF DATA IN COLUMN 1 IS THE INPUT JUST HIT THE RETURN KEY'
WRITE 'OTHERWISE TYPE THE DESIRED COLUMN NUMBER'
LET N1=UNASS.
READ ; N1 INT
DEFAULT N1=1
```

```
"
WRITE 'NAME OF DATA FILE WHERE YOUR OUTPUT IS'
READ F2 NAME
"
WRITE 'IF DATA IN COLUMN 1 IS THE OUTPUT JUST HIT THE RETURN KEY'
WRITE 'OTHERWISE TYPE THE DESIRED COLUMN NUMBER'
LET N2=UNASS.
READ ; N2 INT
DEFAULT N2=1
"
WRITE 'CHOOSE NAME OF FILE FOR THE ESTIMATED SQUARED COHERENCE'
WRITE 'FUNCTION'
READ ZQCOH NAME
"
WRITE 'CHOOSE THE NUMBER OF LAGS IN THE COMPUTATION OF CORRELATION'
WRITE 'FUNCTIONS, I.E. THE WIDTH OF THE TIME WINDOW'
READ NLAG INT
"
SWITCH ECHO ON
"
ASPEC ZQAU<F1(N1) NLAG
ASPEC ZQAY<F2(N2) NLAG
CSPEC ZQCUY<F1(N1) F2(N2) NLAG
"
MOVE ZQCUY1<ZQCUY
SCLOP ZQCUY1(3)<ZQCUY1(3) * -1.
"
FROP ZQCUY1<ZQCUY1*ZQCUY
FROP ZQAUY<ZQAU*ZQAY
FROP ZQCOH<ZQCUY1/ZQAUY
"
MOVE ZQCOH(3)<ZQCOH(2)
"
SWITCH ECHO OFF
"
LET N3=UNASS.
WRITE 'HIT THE RETURN KEY TO PROCEED'
READ ; N3 INT
"
BODE(P) ZQCOH
"
GOTO EXIT
"
"
"
LABEL DFT
"
WRITE 'NAME OF DATA FILE WHERE THE SIGNAL TO BE ANALYSED IS'
READ DATA NAME
"
WRITE 'IF DATA IN COLUMN 1 IS TO BE ANALYSED JUST HIT THE RETURN KEY'
WRITE 'OTHERWISE TYPE THE DESIRED COLUMN NUMBER'
LET N=UNASS.
READ ; N INT
DEFAULT N=1
"
WRITE 'CHOOSE NAME OF FILE FOR THE ESTIMATED SPECTRUM'
READ ZQAS NAME
"
LABEL 3
WRITE 'DO YOU WANT THE AMPLITUDE OR POWER SPECTRUM?'
WRITE 'TYPE AMP OR POW'
READ ZQANS NAME
IF ZQANS EQ AMP GOTO 4
```

```
IF ZQANS EQ POW GOTO 4
GOTO 3
"
LABEL 4
WRITE 'DO YOU WANT TO USE A WINDOW? ANSWER YES OR NO'
READ ZQSW YESNO
IF ZQSW EQ YES GOTO 21
IF ZQSW EQ NO GOTO 22
LABEL 21
LET ZQWND=BH
GOTO 23
LABEL 22
LET ZQWND=BC
LABEL 23
"
DFT (ZQANS) (ZQWND) ZQAS < DATA(N)
"
BODE (AO)ZQAS
"
GOTO EXIT
"
"
"
LABEL TRFDFT
"
WRITE 'NAME OF FILE WHERE YOUR INPUT SIGNAL IS'
READ F1 NAME
"
WRITE 'IF DATA IN COLUMN 1 IS THE INPUT JUST HIT THE RETURN KEY'
WRITE 'OTHERWISE TYPE THE DESIRED COLUMN NUMBER'
LET N1=UNASS.
READ ; N1 INT
DEFAULT N1=1
"
WRITE 'NAME OF DATA FILE WHERE YOUR OUTPUT IS'
READ F2 NAME
"
WRITE 'IF DATA IN COLUMN 1 IS THE OUTPUT JUST HIT THE RETURN KEY'
WRITE 'OTHERWISE TYPE THE DESIRED COLUMN NUMBER'
LET N2=UNASS.
READ ; N2 INT
DEFAULT N2=1
"
WRITE 'CHOOSE NAME OF FILE FOR THE ESTIMATED TRANSFER FUNCTION'
READ ZQTRF NAME
"
WRITE 'DO YOU WANT TO USE A WINDOW? ANSWER YES OR NO'
READ ZQANS YESNO
IF ZQANS EQ YES GOTO 11
IF ZQANS EQ NO GOTO 12
LABEL 11
LET ZQSW=BH
GOTO 13
LABEL 12
LET ZQSW=BC
LABEL 13
"
SWITCH ECHO ON
"
DFT (ZQSW) ZQASU < F1(N1)
DFT (ZQSW) ZQASY < F2(N2)
FROP ZQTRF < ZQASY/ZQASU
"
SWITCH ECHO OFF
```

```
"
LET N3=UNASS.
WRITE 'HIT THE RETURN KEY TO PROCEED'
READ ; N3 INT
"
BODE ZQTRF
"
GOTO EXIT
"
"
"
LABEL PREWHITE
"
LET STDEV.ZQPREWH=1.
"
WRITE 'NAME OF FILE WHERE YOUR INPUT SIGNAL IS'
READ F1 NAME
"
WRITE 'IF DATA IN COLUMN 1 IS THE INPUT JUST HIT THE RETURN KEY'
WRITE 'OTHERWISE TYPE THE DESIRED COLUMN NUMBER'
LET N1=UNASS.
READ ; N1 INT
DEFAULT N1=1
"
WRITE 'NAME OF DATA FILE WHERE YOUR OUTPUT IS'
READ F2 NAME
"
WRITE 'IF DATA IN COLUMN 1 IS THE OUTPUT JUST HIT THE RETURN KEY'
WRITE 'OTHERWISE TYPE THE DESIRED COLUMN NUMBER'
LET N2=UNASS.
READ ; N2 INT
DEFAULT N2=1
"
WRITE 'CHOOSE NAME OF FILE FOR THE PREWHITENED DATA. THE INPUT'
WRITE 'SIGNAL WILL BE PUT IN COLUMN 1 AND THE OUTPUT IN COLUMN 2'
READ ZQPREW NAME
"
WRITE 'CHOOSE ORDER OF PREWHITENING AUTOREGRESSIVE FILTER, MAX 40'
WRITE 'IF YOU JUST HIT THE RETURN KEY YOU GET THE DEFAULT VALUE 10'
READ ; NORDER INT
DEFAULT NORDER=10
"
TURN TEXT OFF
TURN GRAPH OFF
"
STRUC ZQSTRF
NA MAX NORDER
NU MAX 0
X
"
SQR ZQRMAT<F1(N1) ZQSTRF
LS ZQMODEL<ZQSTRF
"
RESID ZQPREW(1)<ZQMODEL F1(N1)
KILL
"
STAT ZQPREW(1) ZQPREWH
SCLOP ZQPREW(1)<ZQPREW(1)/STDEV.ZQPREWH
"
RESID ZQPREW(2)<ZQMODEL F2(N2)
KILL
"
SCLOP ZQPREW(2)<ZQPREW(2)/STDEV.ZQPREWH
"
```

```
DELET ZQMODEL (T)
"
TURN TEXT ON
TURN GRAPH ON
"
GOTO EXIT
"
"
"
LABEL BODE
"
LET F1=UNASS.
LET F2=UNASS.
LET F3=UNASS.
LET F4=UNASS.
LET F5=UNASS.
"
WRITE 'NAME(S) OF DATA FILES TO BE PLOTTED (MAX 5)'
READ F1 NAME ; F2 NAME ; F3 NAME ; F4 NAME ; F5 NAME
"
BODE F1 F2 F3 F4 F5
"
GOTO EXIT
"
"
"
LABEL PLOT
"
LET F1=UNASS.
LET F2=UNASS.
LET F3=UNASS.
LET M1=UNASS.
LET M2=UNASS.
LET M3=UNASS.
LET P1=UNASS.
LET P2=UNASS.
LET P3=UNASS.
LET R1=UNASS.
LET R2=UNASS.
LET R3=UNASS.
LET LP2=UNASS.
LET RP2=UNASS.
LET LP3=UNASS.
LET RP3=UNASS.
"
WRITE 'NAME(S) OF DATA FILES TO BE PLOTTED (MAX 3)'
READ F1 NAME ; F2 NAME ; F3 NAME
"
WRITE 'WHICH COLUMN(S) (MAX 3) OF DATA FILE    ' F1
WRITE 'IF YOU WANT TO PLOT JUST COLUMN 1 HIT THE RETURN KEY'
READ ; M1 INT ; M2 INT ; M3 INT
DEFAULT M1=1
"
IF F2 EQ UNASS. GOTO 5
"
LET LP2=(
LET RP2=)
"
WRITE 'WHICH COLUMN(S) (MAX 3) OF DATA FILE    ' F2
WRITE 'IF YOU WANT TO PLOT JUST COLUMN 1 HIT THE RETURN KEY'
READ ; P1 INT ; P2 INT ; P3 INT
DEFAULT P1=1
"
IF F3 EQ UNASS. GOTO 5
```

```
"
LET LP3=(
LET RP3=)
"
WRITE "WHICH COLUMN(S) (MAX 3) OF DATA FILE    " F3
WRITE "IF YOU WANT TO PLOT JUST COLUMN 1 HIT THE RETURN KEY"
READ ; R1 INT ; R2 INT ; R3 INT
DEFAULT R1=1
"
LABEL 5
WRITE "HOW MANY SAMPLES DO YOU WANT TO PLOT?"
READ NPLX INT
"
PLOT (NPLX) F1(M1 M2 M3) F2 LP2 P1 P2 P3 RP2 F3 LP3 R1 R2 R3 RP3
KILL
"
GOTO EXIT
"
"
"
LABEL EXIT
"
LET I=UNASS.
"
WRITE "HIT THE RETURN KEY TO PROCEED"
WRITE "IF YOU WANT THE EXTENDED DESCRIPTION OF THE POSSIBILITIES TYPE 1"
READ ; I INT
"
IF I EQ 1 GOTO MENU
"
WRITE (FF) " "
WRITE "CHOOSE BETWEEN THE FOLLOWING POSSIBILITIES:"
WRITE " "
WRITE " ACOF  ASPEC  CCOF  CSPEC  TRF  COH  DFT"
WRITE " TRFDFT  PREWHITE  BODE  PLOT  EXIT"
"
GOTO 30
"
"
"
LABEL END
"
"
"
END
```

HJÄLPMACRO´s FÖR CORANA

```
MACRO MACKA ; A
"
"HELP MACRO FOR THE MACRO CORANA
"
LET UNASS.=A
"
END
```

```
MACRO KILL
"
"HELP MACRO FOR THE MACRO CORANA
"
END
```

PREWHITE

Denna MACRO användes för prewhitening (förfiltrering) före t.ex. korskorrelationsanalys. Med hjälp av minsta kvadratidentifiering bestäms ett filter, som approximativt ger vitt brus som utsignal om det matas med den givna insignalen. Sedan filtreras den givna utsignalen genom samma filter. Korskorrelationen beräknas sedan mellan de sålunda filtrerade signalerna. Detta förfaringssätt ger bl.a. en enklare tolkning åt den estimerade korskorrelationsfunktionen, nämligen som en skattning av impulssvaret.

MACRO´n anropas med

>PREWHITE PREW ← INPUT(N1) OUTPUT(N2) ; N



Fig. 1. Illustration av prewhitening.

```
MACRO PREWHITE PREW<INPUT(NCOL1) OUTPUT(NCOL2) ; N
"
"MACRO FOR PREWHITENING OF INPUT AND OUTPUT SIGNALS BEFORE
"CROSSCORRELATION ETC.
"
"INPUT  NAME OF FILE WHERE THE INPUT SIGNAL IS
"NCOL1  COLUMN NUMBER FOR INPUT SIGNAL
"OUTPUT NAME OF FILE WHERE THE OUTPUT SIGNAL IS
"NCOL2  COLUMN NUMBER FOR OUTPUT SIGNAL
"N      ORDER OF AUTOREGRESSIVE FILTER USED FOR THE PRE-
"       WHITENING. THIS PARAMETER IS OPTIONAL. DEFAULT 10
"PREW   NAME OF FILE WHERE THE PREWHITENED SIGNALS ARE PLACED.
"       INPUT SIGNAL IN COLUMN 1, OUTPUT SIGNAL IN COLUMN 2
"
DEFAULT N=10
"
TREND INPUTT<INPUT(NCOL1) 0
TREND OUTPUTT<OUTPUT(NCOL2) 0
"
LET STDEV.ZQPREW=1.
"
TURN TEXT OFF
TURN GRAPH OFF
"
STRUC ZQSTRF
NA MAX N
NU MAX 0
X
SQR ZQRMAT<INPUTT ZQSTRF
LS ZQLSMOD<ZQSTRF
"
RESID PREW(1)<ZQLSMOD INPUTT
KILL
"
STAT PREW(1) ZQPREW
SCLOP PREW(1)<PREW(1)/STDEV.ZQPREW
"
RESID PREW(2)<ZQLSMOD OUTPUTT
KILL
"
SCLOP PREW(2)<PREW(2)/STDEV.ZQPREW
"
DELET ZQLSMOD (T)
"
TURN TEXT ON
TURN GRAPH ON
"
END
```

CCOFF

Detta är en MACRO för beräkning och plottning av korskorrelationsfunktionen för två signaler. Plottning ger en vertikal linje för tidsförskjutningen noll mellan signalerna och ger riktiga skalmarkeringar på horisontella axeln.

MACRO'n anropas med

>CCOFF CC ← DATA1(N1) DATA2(N2) NLAG

```
MACRO CCOFF CC<DATA1(N1) DATA2(N2) NLAG
"
"MACRO FOR COMPUTING AND PLOTTING THE CROSSCORRELATION
"FUNCTION
"
"DATA1   NAME OF FILE WHERE THE INPUT SIGNAL IS
"N1      COLUMN NUMBER FOR INPUT SIGNAL
"DATA2   NAME OF FILE WHERE THE OUTPUT SIGNAL IS
"N2      COLUMN NUMBER FOR OUTPUT SIGNAL
"NLAG    NUMBER OF LAGS
"CC      NAME OF FILE WHERE THE ESTIMATED CROSSCORRELATION
"        FUNCTION IS PLACED
"
LET NS = 2 * NLAG
LET NN = NS + 1
LET NQ = NLAG + 1
"
INSI ZQP NN
LET AMP.=2
LET IFP.=NQ
ZERO
STEP
LET AMP.=1
LET IFP.=1
RAMP
X
"
SCLOP ZQP(2)<ZQP(2) - 1.
SCLOP ZQP(4)<ZQP(1) + 1.
SCLOP ZQP(3)<ZQP(3) - NLAG
"
CCOF CC<DATA1(N1) DATA2(N2) NLAG
"
PLOT (NM) (NN) ZQP(3)<CC ZQP(1 4) (HP) ZQP(2)
"
END
```

LINSPEC

I vissa fall kan det vara lämpligt att beräkna spektrum med linjärt ekvidistanta punkter i stället för med logaritmiskt ekvidistanta punkter, som ASPEC använder. Detta kan göras med LINSPEC, Plottning av det estimerade spektrat kan ske i lin/lin eller log/log skala.

MACRO´n anropas med

>LINSPEC AS ← DATA(NCOL1) DT LSCALE

```
MACRO LINSPEC AS<DATA(NCOL1) DT LSCALE
"
"MACRO FACILITATING COMPUTATION OF AUTOSPECTRUM IN LINEARLY
"EQUIDISTANT POINTS. PLOTTING CAN BE DONE IN LOG/LOG OR
"LIN/LIN SCALES. THE AUTOSPECTRUM IS COMPUTED FOR TWO DECADES.
"
"DATA    NAME OF FILE WHERE THE SIGNAL TO BE ANALYSED IS
"NCOL1   COLUMN NUMBER FOR SIGNAL TO BE ANALYSED
"DT      SAMPLING INTERVAL IN S
"LSCALE  IF LSCALE EQ LIN THEN LIN/LIN SCALES. IF LSCALE EQ
"        LOG THEN LOG/LOG SCALES.
"AS      NAME OF FILE WHERE THE ESTIMATED AUTOSPECTRUM IS PLACED
"
LET WMAX.=3.141593/DT
LET WMIN.=WMAX./100
"
INSI ZSFREQ 100
RAMP WMIN. WMIN.
X
"
ASPEC AS<DATA(NCOL1) 100 ZSFREQ
"
IF LSCALE EQ LIN GOTO ZQS
BODE AS
GOTO EXIT
"
LABEL ZQS
PLOT (100) AS(1)<AS(2)
"
LABEL EXIT
"
END
```

INTSPEC

Denna MACRO beräknar det integrerade spektrat för ett givet autospectrum och plottar det.

MACRO´n anropas med

>INTSPEC ISPEC ← ASPEC

```
MACRO INTSPEC ISPEC<ASPEC
"
"MACRO FOR COMPUTATION OF THE INTEGRATED SPECTRUM GIVEN THE
"AUTOSPECTRUM.
"
"ASPEC   NAME OF FILE WHERE THE AUTOSPECTRUM IS
"ISPEC   NAME OF FILE WHERE THE INTEGRATED SPECTRUM IS PLACED.
"        THE FREQUENCIES ARE IN COLUMN 1 AND THE INTEGRATED
"        SPECTRUM IN COLUMN 3.
"
"NOTICE THAT THE SYSTEM FILES SSSS1 AND SSSS2 ARE NEEDED.
"THE SAMPLING INTERVAL GIVEN IN THESE SYSTEM FILES MUST
"COINCIDE WITH THE SAMPLING INTERVAL OF THE DATA FILE FROM
"WHICH THE AUTOSPECTRUM FILE ASPEC HAS BEEN COMPUTED.
"DEFAULT VALUE 1. S.
"
DETER ISPEC(1)<$SSS1 ASPEC(1)
VECOP ISPEC(2)<ISPEC(1)*ASPEC(2)
DETER ISPEC(3)<$SSS2 ISPEC(2)
MOVE ISPEC(1)<ASPEC(1)
"
LET MAX.ZQISPEC=1.
"
TURN TEXT OFF
STAT ISPEC(3) ZQISPEC
TURN TEXT ON
"
LET MAX.ZQISPEC=MAX.ZQISPEC / 100.
SCLOP ISPEC(3)<ISPEC(3) / MAX.ZQISPEC
"
BODE (P) ISPEC
"
END
```

SYSTEMFILEN SSSS1

```
BEGIN
"
"THE DIFFERENCE EQUATION
"     Y(T) = U(T) - U(T-1)
"IS DESCRIBED IN THIS SYSTEM FILE.
"
DISCRETE MISO TRANSFER FUNCTION
"
SAMPLE INTERVAL 1. S
"
APOLYNOMIAL
1.
"
BPOLYNOMIAL
1. - 1.Q^-1
"
END
```

SYSTEMFILEN SSSS2

```
BEGIN
"
"THE DIFFERENCE EQUATION
"     Y(T) - Y(T-1) = U(T)
"IS DESCRIBED IN THIS SYSTEM FILE.
"
DISCRETE MISO TRANSFER FUNCTION
"
SAMPLE INTERVAL 1. S
"
APOLYNOMIAL
1. - 1.Q^-1
"
BPOLYNOMIAL
1.
"
END
```

TRF

Denna MACRO beräknar en skattning av överföringsfunktionen genom att dividera korspektrum för in- och utsignalerna med autospektrum för insignalen. Överföringsfunktionen plottas också.

MACRO´n anropas med

>TRF TRANSF ← INPUT(NCOL1) OUTPUT(NCOL2) NLAG

```
MACRO TRF TRANSF<INPUT(NCOL1) OUTPUT(NCOL2) NLAG
"
"MACRO FOR ESTIMATION OF THE TRANSFER FUNCTION USING
"SPECTRAL ANALYSIS
"
"INPUT  NAME OF FILE WHERE THE INPUT SIGNAL IS
"NCOL1  COLUMN NUMBER FOR INPUT SIGNAL
"OUTPUT NAME OF FILE WHERE THE OUTPUT SIGNAL IS
"NCOL2  COLUMN NUMBER FOR OUTPUT SIGNAL
"NLAG   NUMBER OF LAGS USED IN THE COMPUTATION OF THE
"       SPECTRA, I.E. THE WIDTH OF THE TIME WINDOW
"TRANSF NAME OF FREQUENCY FILE WHERE THE ESTIMATED
"       TRANSFER FUNCTION IS PLACED
"
ASPEC ZQAS<INPUT(NCOL1) NLAG
CSPEC ZQCS<INPUT(NCOL1) OUTPUT(NCOL2) NLAG
PROP TRANSF<ZQCS/ZQAS
"
BODE TRANSF
"
END
```

COH

Denna MACRO beräknar och plottar med logaritmisk frekvensskala den kvadrerade koherensfunktionen.

MACRO´n anropas med

>COH COHF ← INPUT(NCOL1) OUTPUT(NCOL2) NLAG

```
MACRO COH COHF<INPUT(NCOL1) OUTPUT(NCOL2) NLAG
"
"MACRO FOR COMPUTATION OF THE SQUARED COHERENCE FUNCTION
"
"INPUT  NAME OF FILE WHERE THE INPUT SIGNAL IS
"NCOL1  COLUMN NUMBER FOR INPUT SIGNAL
"OUTPUT NAME OF FILE WHERE THE OUTPUT SIGNAL IS
"NCOL2  COLUMN NUMBER FOR OUTPUT SIGNAL
"NLAG   NUMBER OF LAGS USED IN THE COMPUTATION OF THE
"       SPECTRA, I.E. THE WIDTH OF THE TIME WINDOW
"COHF   NAME OF FILE FOR THE COMPUTED COHERENCE FUNCTION.
"       THE FREQUENCIES ARE IN COLUMN 1, THE COHERENCE
"       FUNCTION IN COLUMNS 2 AND 3, WHICH ARE IDENTICAL
"
ASPEC ZQAU<INPUT(NCOL1) NLAG
ASPEC ZQAY<OUTPUT(NCOL2) NLAG
CSPEC ZQCUY<INPUT(NCOL1) OUTPUT(NCOL2) NLAG
"
MOVE ZQCUY1<ZQCUY
SCLOP ZQCUY1(3)<ZQCUY1(3) * -1.
"
FROP ZQCUY1<ZQCUY * ZQCUY1
FROP ZQAUY<ZQAU*ZQAY
"
FROP COHF<ZQCUY1/ZQAUY
"
MOVE COHF(3)<COHF(2)
"
BODE (P) COHF
"
END
```

DFTT

Denna MACRO skattar spektrum med DFT (Discrete Fourier Transform), och ger möjlighet till utjämning genom att tidsserien delas upp i block för vilka spektra beräknas i tur och ordning. Det resulterande spektrat erhålles sedan genom att medelsvärdebilda över de så skattade spektra. Resultatet plottas.

MACRO'n anropas med

>DFTT SPEC ← DATA(NCOL) NBLOCK NDATA ; SW WND OVERLAP

```
MACRO DFTT C<DATA(NCOL) NBLOCK NDATA ; SW WND OVERLAP
"
"MACRO FOR ESTIMATION OF AUTOSPECTRUM USING DFT
"THE TOTAL DATA SET IS DIVIDED INTO BLOCKS OF LENGTH NBLOCK FOR
"WHICH THE SPECTRA ARE COMPUTED. THE RESULTING SPECTRUM IS THEN
"THE AVERAGE VALUE OF ALL BLOCK SPECTRA.
"
"DATA       NAME OF DATA FILE WHERE THE SIGNAL TO BE ANALYSED IS
"NCOL       COLUMN NUMBER OF SIGNAL TO BE ANALYSED
"NBLOCK     NUMBER OF DATA POINTS PER BLOCK (DETERMINING THE
"           WIDTH OF THE TIME WINDOW). MUST BE AN EVEN NUMBER.
"NDATA      TOTAL NUMBER OF DATA POINTS IN THE FILE DATA
"SW         SW=POW              POWER SPECTRUM COMPUTED (DEFAULT)
"           SW=AMP              AMPLITUDE SPECTRUM COMPUTED
"WND        WND=BC              BOX CAR (I.E. RECTANGULAR) WINDOW
"           WND=BH              BLACKMAN-HARRIS WINDOW (DEFAULT)
"OVERLAP    OVERLAP=YES         50 % OVERLAPPING OF BLOCKS
"           OVERLAP=NO          NO OVERLAPPING (DEFAULT)
"C       NAME OF FREQUENCY FILE FOR THE RESULTING SPECTRUM
"
DEFAULT SW=POW
DEFAULT WND=BH
DEFAULT OVERLAP=NO
"
MOVE B<DATA(NCOL)
"
LET N1=1
LET M=0
"
LABEL 10
LET N2=N1+NBLOCK
LET N2=N2-1
"
IF N2 GT NDATA GOTO 20
"
CUT A<B N1 N2
"
```

```
LET M=M+1
LET NBLOCK1=NBLOCK
IF OVERLAP EQ NO GOTO 5
LET NBLOCK1=NBLOCK/2
LABEL 5
LET N1=N1+NBLOCK1
"
DFT (SW) (WND) SPEC<A
"
IF M GT 1 GOTO 30
"
MOVE C<SPEC
GOTO 10
"
LABEL 30
VECOP C(2)<C(2)+SPEC(2)
GOTO 10
"
LABEL 20
SCLOP C(2)<C(2)/M
"
BODE C
"
END
```

DFTF

Denna MACRO skattar spektrum med DFT (Discrete Fourier Transform) och ger möjlighet till utjämning genom att medelvärdesbildning kan göras över närliggande punkter i frekvensplanet. Resultatet plottas.

MACRO"n anropas med

>DFTF C ← DATA(NCOL) NBLOCK NDATA ; SW WND

```
MACRO DFTF C<DATA(NCOL) NBLOCK NDATA ; SW WND
"
"MACRO FOR ESTIMATION OF AUTOSPECTRUM USING DFT
"THE DISCRETE FOURIER TRANSFORM FOR THE WHOLE DATA SEQUENCE
"IS COMPUTED. THE ESTIMATED SPECTRUM POINTS ARE THEN DIVIDED
"INTO BLOCKS OF LENGTH NBLOCK AND AVERAGED FOR EACH BLOCK.
"THE RESULTING SPECTRUM CONSISTS OF THE AVERAGED POINTS.
"
"DATA      NAME OF DATA FILE WHERE THE SIGNAL TO BE ANALYSED IS
"NCOL      COLUMN NUMBER FOR THE SIGNAL TO BE ANALYSED
"NBLOCK    NUMBER OF FREQUENCY POINTS PER BLOCK (DETERMINING
"          THE WIDTH OF THE FREQUENCY WINDOW)
"NDATA     TOTAL NUMBER OF DATA IN THE DATA FILE DATA
"SW        SW=POW      POWER SPECTRUM ESTIMATED (DEFAULT)
"          SW=AMP      AMPLITUDE SPECTRUM ESTIMATED
"WND       WND=BC      BOX CAR (I.E. RECTANGULAR) WINDOW (DEFAULT)
"          WND=BH      BLACKMAN-HARRIS WINDOW
"C         NAME OF FREQUENCY FILE WHERE THE RESULT IS PLACED
"
DEFAULT SW=POW
DEFAULT WND=BC
"
TURN TEXT OFF
LET SUM.DFT=1.
"
INSI CONST 1
STEP
X
"
DFT (SW) (WND) RAWSPEC<DATA(NCOL)
"
LET N1=1
"
LABEL 10
LET N2=N1+NBLOCK
LET N2=N2-1
LET NDATA1=NDATA/2
"
IF N2 GT NDATA1 GOTO 30
"
CUT A<RAWSPEC N1 N2
"
```

```
STAT A(1) DFT
SCLOP B(1)<CONST * SUM.DFT
STAT A(2) DFT
SCLOP B(2)<CONST * SUM.DFT
SCLOP B(3)<CONST * 0.
"
IF N1 GT 1 GOTO 20
"
MOVE C<B
GOTO 40
"
LABEL 20
CONC C<C B
"
LABEL 40
LET N1=N1+NBLOCK
GOTO 10
"
LABEL 30
TURN TEXT ON
SCLOP C(1)<C(1)/NBLOCK
SCLOP C(2)<C(2)/NBLOCK
"
FHEAD C
9 1
X
"
BODE C
"
END
```

LSID

Avsikten med denna MACRO är att underlätta minsta kvadratidentifiering som annars kräver minst tre kommandon. MACRO anropas med ett kommando som är så likt ML-kommandot som möjligt. Den resulterande modellen skrivs ut på displayet.

MACRO´n anropas med

>LSID MODEL ← DATA(N1 N2) N

```
MACRO LSID MODEL<DATA(N1 N2) N
"
"MACRO FOR ESTIMATION OF A LS MODEL OF ORDER N
"
"DATA   NAME OF FILE WHERE THE INPUT AND OUTPUT SIGNALS ARE
"N1     COLUMN NUMBER FOR INPUT SIGNAL
"N2     COLUMN NUMBER FOR OUTPUT SIGNAL
"N      ORDER OF THE ESTIMATED MODEL
"MODEL  NAME OF FILE WHERE THE ESTIMATED MODEL IS PLACED
"
STRUC ZQSTRF
NA MAX N
NU MAX 1
NB MAX N
KB MAX 1
X
"
SQR ZQRMAT<DATA(N1 N2) ZQSTRF
"
FTEST MODEL
IF FTEST. EQ 0 GOTO 10
DELET MODEL (T)
LABEL 10
"
LS (SC) MODEL<ZQSTRF
SAVE STDEV
X
"
LIST (FF) (T) MODEL
"
END
```

LSORD

Avsikten med denna MACRO är att underlätta vid ordningstalsbestämning i samband med minsta kvadratidentifiering. Tabellen med värden på förlustfunktionen och testkvantiteten AIC (Akaike´s Information Criterion) skrivs ut på displayet för olika ordningstal.

MACRO´n anropas med

>LSORD DATA(N1 N2) NMAX

```
MACRO LSORD DATA(N1 N2) NMAX
"
"MACRO FOR DETERMINING THE ORDER OF A MODEL USING LS-
"IDENTIFICATION
"
"DATA   NAME OF FILE WHERE THE INPUT AND OUTPUT SIGNALS ARE
"N1     COLUMN NUMBER FOR INPUT SIGNAL
"N2     COLUMN NUMBER FOR OUTPUT SIGNAL
"NMAX   MAXIMAL ORDER OF THE MODEL CHECKED
"
STRUC ZQSTRF
NA MAX NMAX
NU MAX 1
NB MAX NMAX
KB MAX 1
NA ACT 1
NB ACT 1
X
"
SQR ZQRMAT<DATA(N1 N2) ZQSTRF
LS ZQMODEL<ZQSTRF
"
DELET ZQMODEL (T)
"
END
```

LSDEL

Avsikten med denna MACRO är att automatisera bestämningen av tidsfördröjningar vid identifiering. Modeller med olika antal tidsfördröjningar skattas med minsta kvadratmetoden. Tabellen med värden på förlustfunktionen och på AIC erhålles på displayet för varierande antal tidsfördröjningar i modellen.

MACRO´n anropas med

>LSDEL DATA(N1 N2) N KMIN KMAX

```
MACRO LSDEL DATA(N1 N2) N KMIN KMAX
"
"MACRO FOR ESTIMATION OF THE PURE TIME DELAY USING LS-
"IDENTIFICATION
"
"DATA   NAME OF FILE WHERE THE INPUT AND OUTPUT SIGNALS ARE
"N1     COLUMN NUMBER FOR INPUT SIGNAL
"N2     COLUMN NUMBER FOR OUTPUT SIGNAL
"N      ORDER OF THE ESTIMATED MODELS
"KMIN   MINIMAL VALUE OF THE TIME DELAY CHECKED
"KMAX   MAXIMAL VALUE OF THE TIME DELAY CHECKED
"
STRUC ZQSTRF
NA MAX N
NU MAX 1
NB MAX N
KB MAX KMAX
KB MIN KMIN
KB ACT KMAX
X
"
SQR ZQRMAT<DATA(N1 N2) ZQSTRF
LS ZQMODEL<ZQSTRF
"
DELET ZQMODEL (T)
"
END
```

FTST

Denna MACRO beräknar testkvantiteten för F-test i samband med ordningstals test. MACRO´n är av fråge/svartyp.

MACRO´n anropas med

>FTST

```
MACRO FTST
"
"MACRO FOR COMPUTATION OF THE VALUE OF THE F-TEST QUANTITY FOR
"TEST OF MODEL ORDER. THE FORMULA USED IS
"    V  - V    N - N
"     1    2         2
"    ******* *******
"       V     N   - N
"         2     2     1
"THE MACRO IS QUESTION AND ANSWER BASED. IT IS ASSUMED THAT
"V1 IS GREATER THAN V2, I.E. N1 MUST BE LESS THAN N2, WHERE
"N1 AND N2 ARE THE NUMBER OF PARAMETERS OF THE MODELS RESP.
"
LABEL 1
WRITE 'VALUES OF LOSS FUNCTIONS, V1 RESP. V2?'
READ V1 NUM V2 NUM
"
LET V1=V1+0.
LET V2=V2+0.
LET DV=V1-V2
IF DV LT 0. GOTO 10
"
LABEL 2
WRITE 'NUMBER OF PARAMETERS IN RESP. MODEL, N1 AND N2?'
READ N1 NUM N2 NUM
"
LET N1=N1+0.
LET N2=N2+0.
LET NP=N2-N1
IF NP LE 0 GOTO 20
"
WRITE 'NUMBER OF SAMPLES?'
READ N NUM
"
LET N=N+0.
"
LET DV=DV/V2
LET NN=N-N2
LET NN=NN/NP
LET DV=DV*NN
"
```

```
WRITE 'F-TEST VALUE:    ' DV
GOTO END
"
LABEL 10
WRITE 'V2 IS GREATER THAN V1'
GOTO 1
"
LABEL 20
WRITE 'N2 MUST BE GREATER THAN N1'
GOTO 2
"
LABEL END
"
END
```

MODANA

Denna MACRO är avsedd för modellanalys. Den testar residualer, beräknar och plottar modellens deterministiska utsignal tillsammans med den verkliga utsignalen och med residualerna. Den beräknar modellens överföringsfunktion och effektspektrum för störningsdelen och plottar dessa. MACRO är avsedd för fallet med en in- och en utsignal.

MACRO´n anropas med

>MODANA MODEL DATA(N1 N2) ; NPLX

```
MACRO MODANA MODEL DATA(N1 N2) ; NPLX
"
"MACRO FOR MODEL ANALYSIS INCLUDING RESIDUAL TESTING, COMPUTA-
"TION AND PLOTTING OF THE DETERMINISTIC MODEL OUTPUT TOGETHER
"WITH THE ACTUAL OUTPUT AND THE RESIDUALS, COMPUTATION OF THE
"TRANSFER FUNCTION OF THE MODEL AND THE POWER SPECTRUM OF THE
"DISTURBANCE MODEL AND PLOTTING OF THESE.
"
"MODEL   NAME OF SYSTEM FILE WHERE THE MODEL TO BE ANALYSED IS
"DATA    NAME OF DATA FILE GENERATING THE MODEL
"N1      COLUMN NUMBER OF INPUT
"N2      COLUMN NUMBER OF OUTPUT
"NPLX    OPTIONAL ARGUMENT (DEFAULT VALUE = THE GLOBAL VARIABLE
"        NPLX.) DETERMINING THE NUMBER OF POINTS PLOTTED PER
"        PAGE
"
"NOTICE THAT WHENEVER A PLOT HAS BEEN FINISHED THE MACRO IS
"WAITING FOR THE COMMAND RESUME TO PROCEED
"
DEFAULT NPLX=NPLX.
"
RESID ZQR1<MODEL DATA(N1 N2) 15 15
SUSPEND
PAGE
SUSPEND
KILL
"
DETER ZQD1<MODEL DATA(N1)
PLOT(NPLX) ZQD1 DATA(N2)/ZQR1
SUSPEND
"
SPTRF ZQTRF<MODEL B/A
BODE ZQTRF
SUSPEND
"
SPTRF (POW) ZQDIST<MODEL C/A
BODE ZQDIST
"
END
```

STEPTEST

Denna MACRO beräknar och plottar stegsvaret för ett givet system.

MACRO´n anropas med

>STEPTEST STEPRESP ← MODEL N

```
MACRO STEPTEST STEPRESP<MODEL N
"
"MACRO GENERATING THE STEP RESPONSE OF A GIVEN SYSTEM
"
"MODEL     NAME OF FILE FOR THE SYSTEM TO BE ANALYSED
"N         NUMBER OF SAMPLES IN THE COMPUTED STEP RESPONSE
"STEPRESP NAME OF FILE FOR THE COMPUTED STEP RESPONSE
"
"NOTICE THAT THE GLOBAL VARIABLE DELTA. MUST BE GIVEN
"THE SAME VALUE AS THE SAMPLING INTERVAL IN THE FILE
"MODEL BEFORE USING THIS MACRO (DEFAULT 1.)
"
INSI ZQSTEP N
STEP
X
"
DETER STEPRESP<MODEL ZQSTEP
"
PLOT (N) STEPRESP
"
END
```

IMPTEST

Denna MACRO beräknar och plottar impulssvaret för ett givet system.

MACRO'n anropas med

>IMPTEST IMPRESP ← MODEL N

```
MACRO IMPTEST IMPRESP<MODEL N
"
"MACRO GENERATING THE IMPULSE RESPONSE OF A GIVEN SYSTEM
"
"MODEL     NAME OF FILE FOR THE SYSTEM TO BE ANALYSED
"N         NUMBER OF SAMPLES IN THE COMPUTED IMPULSE RESPONSE
"IMPRESP   NAME OF FILE FOR THE COMPUTED IMPULSE RESPONSE
"
"NOTICE THAT THE GLOBAL VARIABLE DELTA. MUST BE GIVEN
"THE SAME VALUE AS THE SAMPLING INTERVAL IN THE FILE
"MODEL BEFORE USING THIS MACRO (DEFAULT 1.)
"
INSI ZQPULS N
PULSE
X
"
DETER IMPRESP<MODEL ZQPULS
"
PLOT (N) IMPRESP
"
END
```

RAND

Denna MACRO användes för att göra en Monte Carlo-simulering av en modells tidssvar för en given insignal i akt och mening att testa modellens osäkerhet.

MACRO´n anropas med

>RAND Y ← MOD U NL ; NPLX

```
MACRO RAND Y<MOD U NL ; NPLX
"
"MACRO FOR MONTE CARLO SIMULATION OF TIME RESPONSES IN ORDER TO
"TEST THE UNCERTAINTY OF AN ESTIMATED MODEL.
"
"MOD    SYSTEM FILE WHERE THE ESTIMATED MODEL PARAMETERS AND
"       THEIR ESTIMATED COVARIANCE MATRIX ARE
"U      FILE WITH INPUT SIGNAL FOR THE SIMULATIONS, E.G. A STEP SIGNAL.
"       NOTICE THAT THE SAMPLING INTERVAL OF THIS FILE MUST COINCIDE
"       WITH THE SAMPLING INTERVAL IN THE SYSTEM FILE MOD.
"NL     THE NUMBER OF SIMULATED TIME RESPONSES REQUESTED
"NPLX   OPTIONAL ARGUMENT (DEFAULT VALUE = THE GLOBAL VARIABLE
"       NPLX.) DETERMINING NUMBER OF POINTS PLOTTED PER PAGE,
"       NORMALLY EQUAL TO THE NUMBER OF POINTS IN THE DATA FILE U
"Y      DATA FILE CONTAINING THE SIMULATED TIME RESPONSES AS
"       COLUMNS
"
DEFAULT NPLX=NPLX.
"
FOR I=1 TO NL
RANPA P<MOD
DETER Y(I)<P U
DELET P (T)
NEXT I
"
PLOT (NPLX) Y
"
END
```

PREDERR

Med denna MACRO beräknas och plottas k-stegs prediktionsfelet som funktion av k för en given modell. Det förutsättes att modellen har ett B-polynom och ett C-polynom.

MACRO'n anropas med

>PREDERR SYST ; TSAMP

```
MACRO PREDERR SYST ; TSAMP
"
"MACRO FOR COMPUTATION OF THE K-STEP PREDICTION ERROR AS A
"FUNCTION OF K FOR THE SYSTEM IN THE SYSTEM FILE SYST.
"IT IS ASSUMED THAT THE SYSTEM FILE CONTAINS A B- AND A C-
"POLYNOMIAL
"
"SYST  THE SYSTEM FILE CONTAINING THE SYSTEM TO BE ANALYSED
"TSAMP OPTIONAL ARGUMENT FOR THE SAMPLING INTERVAL (DEFAULT
"      1. S). IT MUST BE EQUAL TO THE SAMPLING INTERVAL IN
"      THE SYSTEM FILE SYST.
"
"NOTICE THAT THE SYSTEM FILE SSSS2 IS NEEDED. THE SAMPLING
"INTERVAL OF SSSS2 MUST COINCIDE WITH THE ONE OF SYST (AND
"THE ARGUMENT TSAMP).
"
DEFAULT TSAMP=1.
"
INSI ZQPULS 20
LET DELTA.=TSAMP
ZERO
PULSE
X
"
DSIM ZQPRED<SYST ZQPULS
VECOP ZQPRED<ZQPRED*ZQPRED
DETER ZQPREDER<SSSS2 ZQPRED
"
PLOT (20) (NL) ZQPREDER
"
END
```

```
SYSTEMFILEN SSSS2


BEGIN
"
"THE DIFFERENCE EQUATION
"     Y(T) - Y(T-1) = U(T)
"IS DESCRIBED IN THIS SYSTEM FILE.
"
DISCRETE MISO TRANSFER FUNCTION
"
SAMPLE INTERVAL 1. S
"
APOLYNOMIAL
1. - 1.Q^-1
"
BPOLYNOMIAL
1.
"
END
```

CODEN: LUTFD2/(TFRT-7170)/1-39/(1979)

NÅGRA MACRO'S FÖR IDPAC

I Gustavsson

Department of Automatic Control
Lund Institute of Technology
August 1979

Lunds universitet
Reprocentralen, 1985

NÅGRA MACRO´S FÖR IDPAC

I Gustavsson

Detta arbete har understötts av Styrelsen för Teknisk Utveckling under anslag nr 78-3763.

SYSTEMFILEN SSSS2

```
BEGIN
"
"THE DIFFERENCE EQUATION
"     Y(T) - Y(T-1) = U(T)
"IS DESCRIBED IN THIS SYSTEM FILE.
"
DISCRETE MISO TRANSFER FUNCTION
"
SAMPLE INTERVAL 1. S
"
APOLYNOMIAL
1. - 1.Q^-1
"
BPOLYNOMIAL
1.
"
END
```


Dokumentutgivare
Lund Institute of Technology
Handläggare Department of Automatic Control
I Gustavsson
Författare
I Gustavsson

Dokumentnamn
REPORT
Utgivningsdatum
August 1979

Dokumentbeteckning
LUTFD2/(TFRT-7170)/1-39/(1979)
Ärendebeteckning

Dokumenttitel och undertitel

Några MACRO´s för IDPAC

(Some MACRO´s for IDPAC)

Referat (sammandrag)

A number of MACRO´s for use in the interactive identification program package IDPAC are presented in this report. They concern the areas of correlation and spectral analysis, least squares identification and model analysis. Some of them are primarily designed for teaching purposes, but most of them are suitable for practical use.

Referat skrivet av
Author
Förslag till ytterligare nyckelord

Klassifikationssystem och -klass(er)

Indextermer (ange källa)

Omfång
39 pages
Övriga bibliografiska uppgifter
Språk
Swedish
Sekretessuppgifter
ISSN
ISBN
Dokumentet kan erhållas från
Department of Automatic Control
Lund Institute of Technology
Box 725, S-220 07 LUND 7, Sweden
Mottagarens uppgifter
Pris

SIS-DB 1
DOKUMENTDATABLAD enligt SIS 62 10 12

Blankett LU 11:25 1976–07

RAND

Denna MACRO användes för att göra en Monte Carlo-simulering av en modells tidssvar för en given insignal i akt och mening att testa modellens osäkerhet.

MACRO´n anropas med

>RAND Y ← MOD U NL ; NPLX

```
MACRO RAND Y<MOD U NL ; NPLX
"
"MACRO FOR MONTE CARLO SIMULATION OF TIME RESPONSES IN ORDER TO
"TEST THE UNCERTAINTY OF AN ESTIMATED MODEL.
"
"MOD    SYSTEM FILE WHERE THE ESTIMATED MODEL PARAMETERS AND
"       THEIR ESTIMATED COVARIANCE MATRIX ARE
"U      FILE WITH INPUT SIGNAL FOR THE SIMULATIONS, E.G. A STEP SIGNAL.
"       NOTICE THAT THE SAMPLING INTERVAL OF THIS FILE MUST COINCIDE
"       WITH THE SAMPLING INTERVAL IN THE SYSTEM FILE MOD.
"NL     THE NUMBER OF SIMULATED TIME RESPONSES REQUESTED
"NPLX   OPTIONAL ARGUMENT (DEFAULT VALUE = THE GLOBAL VARIABLE
"       NPLX.) DETERMINING NUMBER OF POINTS PLOTTED PER PAGE,
"       NORMALLY EQUAL TO THE NUMBER OF POINTS IN THE DATA FILE U
"Y      DATA FILE CONTAINING THE SIMULATED TIME RESPONSES AS
"       COLUMNS
"
DEFAULT NPLX=NPLX.
"
FOR I=1 TO NL
RANPA P<MOD
DETER Y(I)<P U
DELET P (T)
NEXT I
"
PLOT (NPLX) Y
"
END
```


INNEHÅLLSFÖRTECKNING

STEPTEST

Denna MACRO beräknar och plottar stegsvaret för ett givet system.

MACRO´n anropas med

>STEPTEST STEPRESP ← MODEL N

```
MACRO STEPTEST STEPRESP<MODEL N
"
"MACRO GENERATING THE STEP RESPONSE OF A GIVEN SYSTEM
"
"MODEL     NAME OF FILE FOR THE SYSTEM TO BE ANALYSED
"N         NUMBER OF SAMPLES IN THE COMPUTED STEP RESPONSE
"STEPRESP NAME OF FILE FOR THE COMPUTED STEP RESPONSE
"
"NOTICE THAT THE GLOBAL VARIABLE DELTA. MUST BE GIVEN
"THE SAME VALUE AS THE SAMPLING INTERVAL IN THE FILE
"MODEL BEFORE USING THIS MACRO (DEFAULT 1.)
"
INSI ZQSTEP N
STEP
X
"
DETER STEPRESP<MODEL ZQSTEP
"
PLOT (N) STEPRESP
"
END
```


INLEDNING

Denna rapport innehåller ett antal MACRO´s, varav en del, typ CORANA, i första hand är avsedda för undervisningsändamål, medan de flesta är avsedda för praktiska sammanhang. Avsikten är dels att publicera ett antal ofta efterfrågade MACRO´s, typ LSID, som underlättar minsta kvadratidentifiering, men dels också ge exempel på hur användaren själv kan skriva sina egna speciella MACRO´s och därvid utnyttja IDPAC effektivare.

Före varje MACRO ges en mycket kort beskrivning och sedan listas själva MACRO´n med sina kommentarer. Det rekommenderas att användaren läser såväl den inledande beskrivningen som kommentarerna i MACRO´n före användandet.

De MACRO´s som presenteras kan delas in i grupper enligt följande:

Korrelations- och spektralanalys

CORANA	Fråge/svarsstyrt program för korrelations- och spektralanalys
PREWHITE	Förfiltrering (prewhitening)
CCOFF	Beräkning och plottning av korskorrelationsfunktion
LINSPEC	Beräkning och plottning av autospektrum (beräknat för linjärt ekvidistanta punkter)
INTSPEC	Beräkning och plottning av integrerat spectrum
TRF	Skattning och plottning av överföringsfunktion
COH	Beräkning och plottning av koherensfunktion
DFTT	Beräkning, utjämning och plottning av autospektrum (DFT) (utjämning genom uppdelning av tidsserien i mindre block och medelvärdesbildning över de skattade spektra)
DFTF	Beräkning, utjämning och plottning av autospektrum (DFT) (utjämning genom medelvärdesbildning i frekvensplanet)

```
MACRO CORANA
"
"MACRO'S NEEDED
"      MACKA
"      KILL
"
"
"
WRITE (FF) ' '
WRITE 'THIS QUESTION AND ANSWER PROGRAM IS INTENDED TO GUIDE YOU'
WRITE 'THROUGH THE EXERCISE IN CORRELATION ANALYSIS'
WRITE ' '
WRITE ' '
WRITE ' '
WRITE 'HIT THE RETURN KEY TO PROCEED'
"
MACKA
"
LET I=UNASS.
READ ; I INT
"
"
"
LABEL MENU
WRITE (FF) ' '
WRITE 'CHOOSE BETWEEN THE FOLLOWING POSSIBILITIES'
WRITE ' '
WRITE 'ACOF     - COMPUTATION OF THE AUTOCORRELATION FUNCTION'
WRITE 'ASPEC    - COMPUTATION OF THE AUTOSPECTRUM'
WRITE 'CCOF     - COMPUTATION OF THE CROSSCORRELATION FUNCTION'
WRITE 'CSPEC    - COMPUTATION OF THE CROSSPECTRUM'
WRITE 'TRF      - ESTIMATION OF THE TRANSFER FUNCTION'
WRITE 'COH      - ESTIMATION OF THE COHERENCE FUNCTION'
WRITE 'DFT      - ESTIMATION OF SPECTRUM USING DFT'
WRITE 'TRFDFT   - ESTIMATION OF TRANSFER FUNCTION USING DFT'
WRITE 'PREWHITE- PREWHITENING OF DATA FOR CROSSCORRELATION ANALYSIS'
WRITE 'BODE     - PLOTTING OF SPECTRA AND TRANSFER FUNCTIONS'
WRITE 'PLOT     - PLOTTING OF TIME SERIES'
WRITE 'EXIT     - EXIT TO COMMAND MODE'
"
LABEL 30
WRITE ' '
WRITE 'WHAT IS YOUR INTEREST?'
READ ANSWER NAME
"
IF ANSWER EQ ACOF GOTO ACOF
IF ANSWER EQ ASPEC GOTO ASPEC
IF ANSWER EQ CCOF GOTO CCOF
IF ANSWER EQ CSPEC GOTO CSPEC
IF ANSWER EQ TRF GOTO TRF
IF ANSWER EQ COH GOTO COH
IF ANSWER EQ DFT GOTO DFT
IF ANSWER EQ TRFDFT GOTO TRFDFT
IF ANSWER EQ PREWHITE GOTO PREWHITE
IF ANSWER EQ BODE GOTO BODE
IF ANSWER EQ PLOT GOTO PLOT
IF ANSWER EQ EXIT GOTO END
GOTO EXIT
"
"
"
LABEL ACOF
"
WRITE 'NAME OF DATA FILE WHERE THE SIGNAL TO BE ANALYSED IS'
```


FTST

Denna MACRO beräknar testkvantiteten för F-test i samband med ordningstals test. MACRO'n är av fråge/svartyp.

MACRO'n anropas med

>FTST

```
MACRO FTST
"
"MACRO FOR COMPUTATION OF THE VALUE OF THE F-TEST QUANTITY FOR
"TEST OF MODEL ORDER. THE FORMULA USED IS
"    V  - V    N - N
"     1    2        2
"    ******* *******
"       V     N   - N
"        2     2     1
"THE MACRO IS QUESTION AND ANSWER BASED. IT IS ASSUMED THAT
"V1 IS GREATER THAN V2, I.E. N1 MUST BE LESS THAN N2, WHERE
"N1 AND N2 ARE THE NUMBER OF PARAMETERS OF THE MODELS RESP.
"
LABEL 1
WRITE 'VALUES OF LOSS FUNCTIONS, V1 RESP. V2?'
READ V1 NUM V2 NUM
"
LET V1=V1+0.
LET V2=V2+0.
LET DV=V1-V2
IF DV LT 0. GOTO 10
"
LABEL 2
WRITE 'NUMBER OF PARAMETERS IN RESP. MODEL, N1 AND N2?'
READ N1 NUM N2 NUM
"
LET N1=N1+0.
LET N2=N2+0.
LET NP=N2-N1
IF NP LE 0 GOTO 20
"
WRITE 'NUMBER OF SAMPLES?'
READ N NUM
"
LET N=N+0.
"
LET DV=DV/V2
LET NN=N-N2
LET NN=NN/NP
LET DV=DV*NN
"
```

```
WRITE 'OTHERWISE TYPE THE DESIRED COLUMN NUMBER'
LET N2=UNASS.
READ ; N2 INT
DEFAULT N2=1
"
WRITE 'CHOOSE NAME OF FILE FOR THE ESTIMATED CORRELATION FUNCTION'
READ ZQCC NAME
"
WRITE 'CHOOSE THE MAXIMUM NUMBER OF LAGS IN THE COMPUTATION OF'
WRITE 'THE CROSSCORRELATION FUNCTION'
READ NLAG INT
"
LET NS = 2 * NLAG
LET NN = NS + 1
LET NQ = NLAG + 1
"
INSI ZQP NN
LET AMP.=2.
LET IFP.=NQ
ZERO
STEP
LET AMP.=1.
LET IFP.=1
RAMP
X
"
SCLOP ZQP(2)<ZQP(2) - 1.
SCLOP ZQP(4)<ZQP(1) + 1.
SCLOP ZQP(3)<ZQP(3) - NLAG
"
CCOF ZQCC<F1(N1) F2(N2) NLAG
"
PLOT (NM) (NN) ZQP(3) < ZQCC ZQP(1 4) (HP) ZQP(2)
"
GOTO EXIT
"
"
"
LABEL CSPEC
"
WRITE 'NAME OF DATA FILE WHERE YOUR INPUT SIGNAL IS'
READ F1 NAME
"
WRITE 'IF DATA IN COLUMN 1 IS THE INPUT JUST HIT THE RETURN KEY'
WRITE 'OTHERWISE TYPE THE DESIRED COLUMN NUMBER'
LET N1=UNASS.
READ ; N1 INT
DEFAULT N1=1
"
WRITE 'NAME OF DATA FILE WHERE YOUR OUTPUT SIGNAL IS'
READ F2 NAME
"
WRITE 'IF DATA IN  COLUMN 1 IS THE OUTPUT JUST HIT THE RETURN KEY'
WRITE 'OTHERWISE TYPE THE DESIRED COLUMN NUMBER'
LET N2=UNASS.
READ ; N2 INT
DEFAULT N2=1
"
WRITE 'CHOOSE NAME OF FILE FOR THE ESTIMATED CROSSPECTRUM'
READ ZQCS NAME
"
WRITE 'CHOOSE THE NUMBER OF LAGS IN THE COMPUTATION OF THE CORRELATION'
WRITE 'FUNCTIONS, I.E. THE WIDTH OF THE TIME WINDOW'
READ NLAG INT
```


LSORD

Avsikten med denna MACRO är att underlätta vid ordningstalsbestämning i samband med minsta kvadratidentifiering. Tabellen med värden på förlustfunktionen och testkvantiteten AIC (Akaike's Information Criterion) skrivs ut på displayet för olika ordningstal.

MACRO'n anropas med

>LSORD DATA(N1 N2) NMAX

```
MACRO LSORD DATA(N1 N2) NMAX
"
"MACRO FOR DETERMINING THE ORDER OF A MODEL USING LS-
"IDENTIFICATION
"
"DATA  NAME OF FILE WHERE THE INPUT AND OUTPUT SIGNALS ARE
"N1    COLUMN NUMBER FOR INPUT SIGNAL
"N2    COLUMN NUMBER FOR OUTPUT SIGNAL
"NMAX  MAXIMAL ORDER OF THE MODEL CHECKED
"
STRUC ZQSTRF
NA MAX NMAX
NU MAX 1
NB MAX NMAX
KB MAX 1
NA ACT 1
NB ACT 1
X
"
SQR ZQRMAT<DATA(N1 N2) ZQSTRF
LS ZQMODEL<ZQSTRF
"
DELET ZQMODEL (T)
"
END
```

```
"
LET N3=UNASS.
WRITE 'HIT THE RETURN KEY TO PROCEED'
READ ; N3 INT
"
BODE ZQTRF
"
GOTO EXIT
"
"
"
LABEL PREWHITE
"
LET STDEV.ZQPREWH=1.
"
WRITE 'NAME OF FILE WHERE YOUR INPUT SIGNAL IS'
READ F1 NAME
"
WRITE 'IF DATA IN COLUMN 1 IS THE INPUT JUST HIT THE RETURN KEY'
WRITE 'OTHERWISE TYPE THE DESIRED COLUMN NUMBER'
LET N1=UNASS.
READ ; N1 INT
DEFAULT N1=1
"
WRITE 'NAME OF DATA FILE WHERE YOUR OUTPUT IS'
READ F2 NAME
"
WRITE 'IF DATA IN COLUMN 1 IS THE OUTPUT JUST HIT THE RETURN KEY'
WRITE 'OTHERWISE TYPE THE DESIRED COLUMN NUMBER'
LET N2=UNASS.
READ ; N2 INT
DEFAULT N2=1
"
WRITE 'CHOOSE NAME OF FILE FOR THE PREWHITENED DATA. THE INPUT'
WRITE 'SIGNAL WILL BE PUT IN COLUMN 1 AND THE OUTPUT IN COLUMN 2'
READ ZQPREW NAME
"
WRITE 'CHOOSE ORDER OF PREWHITENING AUTOREGRESSIVE FILTER, MAX 40'
WRITE 'IF YOU JUST HIT THE RETURN KEY YOU GET THE DEFAULT VALUE 10'
READ ; NORDER INT
DEFAULT NORDER=10
"
TURN TEXT OFF
TURN GRAPH OFF
"
STRUC ZQSTRF
NA MAX NORDER
NU MAX 0
X
"
SQR ZQRMAT<F1(N1) ZQSTRF
LS ZQMODEL<ZQSTRF
"
RESID ZQPREW(1)<ZQMODEL F1(N1)
KILL
"
STAT ZQPREW(1) ZQPREWH
SCLOP ZQPREW(1)<ZQPREW(1)/STDEV.ZQPREWH
"
RESID ZQPREW(2)<ZQMODEL F2(N2)
KILL
"
SCLOP ZQPREW(2)<ZQPREW(2)/STDEV.ZQPREWH
"
```



```
LET M=M+1
LET NBLOCK1=NBLOCK
IF OVERLAP EQ NO GOTO 5
LET NBLOCK1=NBLOCK/2
LABEL 5
LET N1=N1+NBLOCK1
"
DFT (SW) (WND) SPEC<A
"
IF M GT 1 GOTO 30
"
MOVE C<SPEC
GOTO 10
"
LABEL 30
VECOP C(2)<C(2)+SPEC(2)
GOTO 10
"
LABEL 20
SCLOP C(2)<C(2)/M
"
BODE C
"
END
```

DFTT

Denna MACRO skattar spektrum med DFT (Discrete Fourier Transform), och ger möjlighet till utjämning genom att tidsserien delas upp i block för vilka spektra beräknas i tur och ordning. Det resulterande spektrat erhålles sedan genom att medelsvärdebilda över de så skattade spektra. Resultatet plottas.

MACRO'n anropas med

>DFTT SPEC ← DATA(NCOL) NBLOCK NDATA ; SW WND OVERLAP

```
MACRO DFTT C<DATA(NCOL) NBLOCK NDATA ; SW WND OVERLAP
"
"MACRO FOR ESTIMATION OF AUTOSPECTRUM USING DFT
"THE TOTAL DATA SET IS DIVIDED INTO BLOCKS OF LENGTH NBLOCK FOR
"WHICH THE SPECTRA ARE COMPUTED. THE RESULTING SPECTRUM IS THEN
"THE AVERAGE VALUE OF ALL BLOCK SPECTRA.
"
"DATA       NAME OF DATA FILE WHERE THE SIGNAL TO BE ANALYSED IS
"NCOL       COLUMN NUMBER OF SIGNAL TO BE ANALYSED
"NBLOCK     NUMBER OF DATA POINTS PER BLOCK (DETERMINING THE
"           WIDTH OF THE TIME WINDOW). MUST BE AN EVEN NUMBER.
"NDATA      TOTAL NUMBER OF DATA POINTS IN THE FILE DATA
"SW         SW=POW              POWER SPECTRUM COMPUTED (DEFAULT)
"           SW=AMP              AMPLITUDE SPECTRUM COMPUTED
"WND        WND=BC              BOX CAR (I.E. RECTANGULAR) WINDOW
"           WND=BH              BLACKMAN-HARRIS WINDOW (DEFAULT)
"OVERLAP    OVERLAP=YES         50 % OVERLAPPING OF BLOCKS
"           OVERLAP=NO          NO OVERLAPPING (DEFAULT)
"C       NAME OF FREQUENCY FILE FOR THE RESULTING SPECTRUM
"
DEFAULT SW=POW
DEFAULT WND=BH
DEFAULT OVERLAP=NO
"
MOVE B<DATA(NCOL)
"
LET N1=1
LET M=0
"
LABEL 10
LET N2=N1+NBLOCK
LET N2=N2-1
"
IF N2 GT NDATA GOTO 20
"
CUT A<B N1 N2
"
```



```
DELET ZQMODEL (T)
"
TURN TEXT ON
TURN GRAPH ON
"
GOTO EXIT
"
"
"
LABEL BODE
"
LET F1=UNASS.
LET F2=UNASS.
LET F3=UNASS.
LET F4=UNASS.
LET F5=UNASS.
"
WRITE 'NAME(S) OF DATA FILES TO BE PLOTTED (MAX 5)'
READ F1 NAME ; F2 NAME ; F3 NAME ; F4 NAME ; F5 NAME
"
BODE F1 F2 F3 F4 F5
"
GOTO EXIT
"
"
"
LABEL PLOT
"
LET F1=UNASS.
LET F2=UNASS.
LET F3=UNASS.
LET M1=UNASS.
LET M2=UNASS.
LET M3=UNASS.
LET P1=UNASS.
LET P2=UNASS.
LET P3=UNASS.
LET R1=UNASS.
LET R2=UNASS.
LET R3=UNASS.
LET LP2=UNASS.
LET RP2=UNASS.
LET LP3=UNASS.
LET RP3=UNASS.
"
WRITE 'NAME(S) OF DATA FILES TO BE PLOTTED (MAX 3)'
READ F1 NAME ; F2 NAME ; F3 NAME
"
WRITE 'WHICH COLUMN(S) (MAX 3) OF DATA FILE   ' F1
WRITE 'IF YOU WANT TO PLOT JUST COLUMN 1 HIT THE RETURN KEY'
READ ; M1 INT ; M2 INT ; M3 INT
DEFAULT M1=1
"
IF F2 EQ UNASS. GOTO 5
"
LET LP2=(
LET RP2=)
"
WRITE 'WHICH COLUMN(S) (MAX 3) OF DATA FILE   ' F2
WRITE 'IF YOU WANT TO PLOT JUST COLUMN 1 HIT THE RETURN KEY'
READ ; P1 INT ; P2 INT ; P3 INT
DEFAULT P1=1
"
IF F3 EQ UNASS. GOTO 5
```

TRF

Denna MACRO beräknar en skattning av överföringsfunktionen genom att dividera korspektrum för in- och utsignalerna med autospektrum för insignalen. Överföringsfunktionen plottas också.

MACRO´n anropas med

>TRF TRANSF ← INPUT(NCOL1) OUTPUT(NCOL2) NLAG

```
MACRO TRF TRANSF<INPUT(NCOL1) OUTPUT(NCOL2) NLAG
"
"MACRO FOR ESTIMATION OF THE TRANSFER FUNCTION USING
"SPECTRAL ANALYSIS
"
"INPUT   NAME OF FILE WHERE THE INPUT SIGNAL IS
"NCOL1   COLUMN NUMBER FOR INPUT SIGNAL
"OUTPUT  NAME OF FILE WHERE THE OUTPUT SIGNAL IS
"NCOL2   COLUMN NUMBER FOR OUTPUT SIGNAL
"NLAG    NUMBER OF LAGS USED IN THE COMPUTATION OF THE
"        SPECTRA, I.E. THE WIDTH OF THE TIME WINDOW
"TRANSF  NAME OF FREQUENCY FILE WHERE THE ESTIMATED
"        TRANSFER FUNCTION IS PLACED
"
ASPEC ZQAS<INPUT(NCOL1) NLAG
CSPEC ZQCS<INPUT(NCOL1) OUTPUT(NCOL2) NLAG
FROP TRANSF<ZQCS/ZQAS
"
BODE TRANSF
"
END
```


HJÄLPMACRO´s FÖR CORANA

```
MACRO MACKA ; A
"
"HELP MACRO FOR THE MACRO CORANA
"
LET UNASS.=A
"
END
```

```
MACRO KILL
"
"HELP MACRO FOR THE MACRO CORANA
"
END
```

CCOFF

Detta är en MACRO för beräkning och plottning av korskorrelationsfunktionen för två signaler. Plottning ger en vertikal linje för tidsförskjutningen noll mellan signalerna och ger riktiga skalmarkeringar på horisontella axeln.

MACRO´n anropas med

>CCOFF CC ← DATA1(N1) DATA2(N2) NLAG

```
MACRO CCOFF CC<DATA1(N1) DATA2(N2) NLAG
"
"MACRO FOR COMPUTING AND PLOTTING THE CROSSCORRELATION
"FUNCTION
"
"DATA1   NAME OF FILE WHERE THE INPUT SIGNAL IS
"N1      COLUMN NUMBER FOR INPUT SIGNAL
"DATA2   NAME OF FILE WHERE THE OUTPUT SIGNAL IS
"N2      COLUMN NUMBER FOR OUTPUT SIGNAL
"NLAG    NUMBER OF LAGS
"CC      NAME OF FILE WHERE THE ESTIMATED CROSSCORRELATION
"        FUNCTION IS PLACED
"
LET NS = 2 * NLAG
LET NN = NS + 1
LET NQ = NLAG + 1
"
INSI ZQP NN
LET AMP.=2
LET IFP.=NQ
ZERO
STEP
LET AMP.=1
LET IFP.=1
RAMP
X
"
SCLOP ZQP(2)<ZQP(2) - 1.
SCLOP ZQP(4)<ZQP(1) + 1.
SCLOP ZQP(3)<ZQP(3) - NLAG
"
CCOF CC<DATA1(N1) DATA2(N2) NLAG
"
PLOT (NM) (NN) ZQP(3)<CC ZQP(1 4) (HP) ZQP(2)
"
END
```


LINSPEC

I vissa fall kan det vara lämpligt att beräkna spektrum med linjärt ekvidistanta punkter i stället för med logaritmiskt ekvidistanta punkter, som ASPEC använder. Detta kan göras med LINSPEC, Plottning av det estimerade spektrat kan ske i lin/lin eller log/log skala.

MACRO´n anropas med

>LINSPEC AS ← DATA(NCOL1) DT LSCALE

```
MACRO LINSPEC AS<DATA(NCOL1) DT LSCALE
"
"MACRO FACILITATING COMPUTATION OF AUTOSPECTRUM IN LINEARLY
"EQUIDISTANT POINTS. PLOTTING CAN BE DONE IN LOG/LOG OR
"LIN/LIN SCALES. THE AUTOSPECTRUM IS COMPUTED FOR TWO DECADES.
"
"DATA    NAME OF FILE WHERE THE SIGNAL TO BE ANALYSED IS
"NCOL1   COLUMN NUMBER FOR SIGNAL TO BE ANALYSED
"DT      SAMPLING INTERVAL IN S
"LSCALE  IF LSCALE EQ LIN THEN LIN/LIN SCALES. IF LSCALE EQ
"        LOG THEN LOG/LOG SCALES.
"AS      NAME OF FILE WHERE THE ESTIMATED AUTOSPECTRUM IS PLACED
"
LET WMAX.=3.141593/DT
LET WMIN.=WMAX./100
"
INSI ZSFREQ 100
RAMP WMIN. WMIN.
X
"
ASPEC AS<DATA(NCOL1) 100 ZSFREQ
"
IF LSCALE EQ LIN GOTO ZQS
BODE AS
GOTO EXIT
"
LABEL ZQS
PLOT (100) AS(1)<AS(2)
"
LABEL EXIT
"
END
```